Title

stata.com

qreg postestimation — Postestimation tools for qreg, iqreg, sqreg, and bsqreg

Postestimation commands

The following postestimation command is of special interest after `sqreg`:

Command	Description
`estat coefplot`	plot coefficients and their confidence intervals at different quantiles

The following postestimation commands are available after `qreg`, `iqreg`, `bsqreg`, and `sqreg`:

Command	Description
`contrast`	contrasts and ANOVA-style joint tests of estimates
`estat summarize`	summary statistics for the estimation sample
`estat vce`	variance–covariance matrix of the estimators (VCE)
`estimates`	cataloging estimation results
`etable`	table of estimation results
†`forecast`	dynamic forecasts and simulations
*`hausman`	Hausman's specification test
`lincom`	point estimates, standard errors, testing, and inference for linear combinations of coefficients
`linktest`	link test for model specification
`margins`	marginal means, predictive margins, marginal effects, and average marginal effects
`marginsplot`	graph the results from margins (profile plots, interaction plots, etc.)
`nlcom`	point estimates, standard errors, testing, and inference for nonlinear combinations of coefficients
`predict`	predictions and their SEs, residuals, etc.
`predictnl`	point estimates, standard errors, testing, and inference for generalized predictions
`pwcompare`	pairwise comparisons of estimates
`test`	Wald tests of simple and composite linear hypotheses
`testnl`	Wald tests of nonlinear hypotheses

*`hausman` is not appropriate after `bsqreg`, `isqreg`, or `sqreg`.

†`forecast` is not appropriate with `mi` estimation results.

predict

Description for predict

`predict` creates a new variable containing predictions such as linear predictions, standard errors, and residuals.

Menu for predict

Statistics > Postestimation

Syntax for predict

For qreg, iqreg, and bsqreg

```
predict [type] newvar [if] [in] [, [xb|stdp|residuals]]
```

For sqreg

```
predict [type] newvar [if] [in] [, equation(eqno[,eqno]) statistic]
```

statistic	Description
Main	
`xb`	linear prediction; the default
`stdp`	standard error of the linear prediction
`stddp`	standard error of the difference in linear predictions
`residuals`	residuals

These statistics are available both in and out of sample; type `predict ... if e(sample) ...` if wanted only for the estimation sample.

Options for predict

Main

`xb`, the default, calculates the linear prediction.

`stdp` calculates the standard error of the linear prediction.

`stddp` is allowed only after you have fit a model using `sqreg`. The standard error of the difference in linear predictions ($\mathbf{x}_{1j}\mathbf{b} - \mathbf{x}_{2j}\mathbf{b}$) between equations 1 and 2 is calculated.

`residuals` calculates the residuals, that is, $y_j - \mathbf{x}_j\mathbf{b}$.

`equation(`*eqno*`[,`*eqno*`])` specifies the equation to which you are making the calculation.

`equation()` is filled in with one *eqno* for the `xb`, `stdp`, and `residuals` options. `equation(#1)` would mean that the calculation is to be made for the first equation, `equation(#2)` would mean the second, and so on. You could also refer to the equations by their names. `equation(income)` would refer to the equation named income and `equation(hours)` to the equation named hours.

If you do not specify `equation()`, results are the same as if you had specified `equation(#1)`.

To use `stddp`, you must specify two equations. You might specify `equation(#1, #2)` or `equation(q80, q20)` to indicate the 80th and 20th quantiles.

margins

Description for margins

`margins` estimates margins of response for linear predictions.

Menu for margins

Statistics > Postestimation

Syntax for margins

`margins` [*marginlist*] [`,` *options*]

`margins` [*marginlist*] `,` `predict(`*statistic* ...`)` [*options*]

statistic	Description
`xb`	linear prediction; the default
`stdp`	not allowed with `margins`
`stddp`	not allowed with `margins`
`residuals`	not allowed with `margins`

Statistics not allowed with `margins` are functions of stochastic quantities other than `e(b)`.

For the full syntax, see [R] **margins**.

estat

Description for estat

`estat coefplot` plots the estimated coefficients and their confidence intervals (CIs) after `sqreg`.

Menu for estat

Statistics > Postestimation

Syntax for estat

```
estat coefplot [varname] [, options]
```

varname is one of the variables specified when fitting `sqreg`; the default is the first endogenous variable.

options	Description
`noci`	do not plot the CIs
`nools`	do not plot the ordinary least-squares (OLS) estimates
Plot	
connect_options	change look of lines or connecting method
marker_options	change look of markers (color, size, etc.)
CI plot	
`ciopts(`*area_options*`)`	affect rendition of the pointwise CIs
Line options	
`lineopts(`*cline_options*`)`	affect rendition of reference line identifying the OLS estimates
Y axis, X axis, Titles, Legend, Overall	
twoway_options	any options other than `by()` documented in [G-3] ***twoway_options***

Options for estat

`noci` removes plots of the pointwise CIs. The default is to plot the CIs.

`nools` removes the plot of the OLS estimates. The default is to plot the OLS reference line.

Plot

connect_options specify how points on a graph are to be connected; [G-3] ***connect_options***.

marker_options affect the rendition of markers drawn at the plotted points, including their shape, size, color, and outline; see [G-3] ***marker_options***.

CI plot

`ciopts(`*area_options*`)` affects rendition of the pointwise CIs; see [G-3] ***area_options***.

Line options

`lineopts(`*cline_options*`)` affects rendition of reference line identifying the OLS estimates; see [G-3] ***cline_options***.

Y axis, X axis, Titles, Legend, Overall

twoway_options are any of the options documented in [G-3] ***twoway_options***, excluding `by()`. These include options for titling the graph (see [G-3] ***title_options***) and for saving the graph to disk (see [G-3] ***saving_option***).

Remarks and examples

stata.com

▷ Example 1: Predictions after qreg and iqreg

In example 4 of [R] **qreg**, we fit regressions for the lower and the upper quartile of the `price` variable. The `predict` command can be used to obtain the linear prediction after each regression.

```
. use https://www.stata-press.com/data/r18/auto
(1978 automobile data)
. qreg price weight length foreign, quantile(.25)
  (output omitted)
. predict q25
(option xb assumed; fitted values)
. qreg price weight length foreign, quantile(.75)
  (output omitted)
. predict q75
(option xb assumed; fitted values)
```

We can use the variables generated by `predict` to compute the predicted interquartile range, that is,

```
. generate iqr1 = q75 - q25
```

If we directly perform the interquartile range regression with the `iqreg` command, we can predict the interquartile range and also the standard error for the prediction.

```
. iqreg price weight length foreign, quantile(.25 .75)
  (output omitted)
. predict iqr2
(option xb assumed; fitted values)
. predict stdp, stdp
```

We now plot the predicted interquartile range versus variable length:

```
. scatter iqr2 length
```



As stated in example 5 of [R] **qreg**, the negative coefficient for the `length` variable means that increases in length imply decreases in the interquartile range and therefore in price dispersion. Consequently, we could have expected a downward trend in the plot, but there is not. This is because the regression output indicates that when we hold the rest of the variables constant, an increase in `length` leads to a decrease in `iqr2`. However, there is a high correlation between `weight` and `length`, which could be masking the effect of `length` on `iqr2`. We can achieve a better visualization by using a contour plot.

```
. twoway contour iqr2 weight length, level(10)
```



We can see the effect by setting a fixed value of `length` on the vertical axis, say, 3,000 lbs. When we move from left to right on the horizontal axis, we see that for small values of `length`, `iqr2` values are shown in red, meaning high values, and when we move toward the right, the graph indicates transition into increasingly smaller values.

◁

▷ Example 2: Coefficients plot after sqreg

In example 5 of [R] **qreg**, we simultaneously estimated the quantile regressions at different quantile indexes.

```
. use https://www.stata-press.com/data/r18/auto, clear
(1978 automobile data)
. set seed 1001
. sqreg price weight length foreign, q(.25 .5 .75) reps(100)
```

(*output omitted*)

We can now use `estat coefplot` to visualize the coefficients at different quantiles. For example, we can type `estat coefplot weight` to see the trend of the effects of `weight` on `price` across quantiles.

```
. estat coefplot weight
```

The resulting graph shows that there is an upward trend in the effects. For reference, the red line shows the OLS estimates.

◁

Also see

[R] **qreg** — Quantile regression

[U] **20 Estimation and postestimation commands**